

Appendix A

Java Modeling Language Reference

Daniel Grahl

This appendix serves as a comprehensive reference for the syntax and semantics the dialect of the Java Modeling Language (JML) that is used in the KeY system, version 2.6. The extensions for information flow introduced in Section 13.4 are also included. Section A.1 presents the full syntax of JML as it is supported by the KeY system. In Section A.2, the semantics of JML expressions is given through a translation to JavaDL. Refer to Chapters 7ff. for in depth explanations and a discussion on these items. Contract semantics are not covered here; they are treated extensively in Section 8.2. We cover the issue of well-definedness in Section A.3.

A.1 JML Syntax

The syntax of JML is heavily intertwined with the syntax of the Java language. The JML reference manual [Leavens et al., 2013] presents a complete grammar for Java enriched with (standard) JML specifications. This includes every grammatical feature—from lexical tokens to compilation units. Here, in this appendix, we take on another approach in not reiterating the entire Java syntax, but indicating where the given definitions would be injected into it. This section only describes syntax; we assume that it is clear from the context how to produce expressions that are well-typed; in doubt, refer to the JML reference manual [Leavens et al., 2013].

Since JML has been designed for other analysis approaches, not all constructs that are introduced in the JML reference manual [Leavens et al., 2013] do make sense for the KeY approach. We present only that part of the syntax for which we can give a formal semantics. Please note that the targeted subset of Java does not include all Java 5–8 features, such as autoboxing, generics, etc. We also omit the JML delimiters `//@` (end of line style) and `/*@ */` (block style). We structure this section after the different locations where JML specifications may appear.

The grammar is given in a Backus Naur style using the postfix operators [?] (optional occurrence), * (any number of occurrences), and ⁺ (at least one occurrence).

Table A.1 Additional class members in JML

— JML Syntax —

```

⟨ClassElem⟩ ::= ⟨ClassSpec⟩ | ⟨MMthd⟩
⟨ClassSpec⟩ ::= ⟨Visibility⟩? ( ⟨ClassInv⟩ | ⟨FieldDecl⟩ | ⟨Represents⟩ | ⟨MAccess⟩ );
⟨ClassInv⟩ ::= ⟨StaticOrInstance⟩? ⟨ClassInvKw⟩ ⟨BoolExpr⟩
⟨StaticOrInstance⟩ ::= static | instance
⟨ClassInvKw⟩ ::= invariant | constraint | initially | axiom
⟨FieldDecl⟩ ::= ( ghost | model ) ⟨Type⟩ ⟨Id⟩+
⟨Represents⟩ ::= represents ⟨Id⟩ = ⟨Expr⟩ | represents ⟨Id⟩ \such_that ⟨BoolExpr⟩
⟨MAccess⟩ ::= accessible ⟨Id⟩ : ⟨LocSetExpr⟩ ⟨Mby⟩?
⟨MMthd⟩ ::= ⟨Contract⟩ ( no_state | two_state )? model ⟨Type⟩ ⟨Id⟩ ( ⟨Params⟩ )
    { return ⟨Expr⟩; }
⟨Params⟩ ::= ( ⟨Type⟩ ⟨Id⟩ ( , ⟨Type⟩ ⟨Id⟩ )* )?

```

— JML Syntax —

In Table A.1, the rule $\langle \text{ClassElem} \rangle$ refers to elements which may additionally appear as a Java class member (see [Gosling et al., 2013, Sect. 8.2]). Rule $\langle \text{Id} \rangle$ refers to valid identifiers in Java (see *ibid.*, Sect. 6) for types (i.e., classes or interfaces), variables, fields, and methods.

Table A.2 Contract grammar in JML

— JML Syntax —

```

⟨Contract⟩ ::= also? ⟨SpecCase⟩ ( also ⟨SpecCase⟩ )*
⟨SpecCase⟩ ::= ⟨Visibility⟩? ⟨Behavior⟩? ⟨Clause⟩*
⟨Visibility⟩ ::= public | protected | private
⟨Behavior⟩ ::= normal_behavior | normal_behaviour |
    exceptional_behavior | exceptional_behaviour
⟨Clause⟩ ::= ( ⟨Requires⟩ | ⟨Ensures⟩ | ⟨Signals⟩ | ⟨SignalsOnly⟩ | ⟨Diverges⟩ |
    ⟨Determins⟩ | ⟨Assign⟩ | ⟨Acc⟩ | ⟨Mby⟩ ); | { | ⟨Clause⟩* | }
⟨Requires⟩ ::= ( requires | pre ) ⟨BoolExpr⟩
⟨Ensures⟩ ::= ( ensures | post ) ⟨BoolExpr⟩
⟨Signals⟩ ::= ( signals | ensures ) ⟨Type⟩ ⟨Id⟩? ⟨BoolExpr⟩
⟨SignalsOnly⟩ ::= ⟨Type⟩ ( , ⟨Type⟩ )* | \nothing | \everything
⟨Diverges⟩ ::= diverges ⟨BoolExpr⟩
⟨Determins⟩ ::= determines ( ⟨Exprs⟩ | \nothing ) \by ( ⟨Exprs⟩ | \itself | \nothing )
    ( \declassifies ⟨Exprs⟩ | erases ⟨Exprs⟩ )? ( \new_objects ⟨Exprs⟩ )?
⟨Assign⟩ ::= ⟨AssignKw⟩ ( ⟨LocSetExpr⟩ ( , ⟨LocSetExpr⟩ )* |
    \nothing | \strictly_nothing | \everything )
⟨AssignKw⟩ ::= assignable | modifiable | modifies
⟨Acc⟩ ::= accessible ( ⟨LocSetExpr⟩ ( , ⟨LocSetExpr⟩ )* | \nothing | \everything )
⟨Mby⟩ ::= measured_by ⟨Exprs⟩

```

— JML Syntax —

Table A.2 shows the grammar used for method contracts. Contracts can appear immediately before method declarations (modulo whitespace; see *ibid.*, Sect. 8.4).

The standard JML clauses interpreted by KeY are limited to **requires**, **ensures**, **signals**, **signals_only**, **diverges**, **assignable**, **accessible**, **measured_by**. For all of these, their synonyms are defined, too (not all are shown here). Other clauses can still be parsed in KeY, but will be ignored. Additionally, the **determines** clause (see Section 13.4) extends standard JML.

Table A.3 shows modifiers in addition to Java’s modifiers (see *ibid.*, Sects. 8.1.1, 8.3.1, 8.4.3, 9.1.1). Note that, technically, the **nullable** and **non_null** are also modifiers, but for more readable semantics we treat them as type information.

Table A.3 Modifiers in JML

— JML Syntax —

$\langle Mod \rangle ::= \text{pure} \mid \text{strictly_pure} \mid \text{helper} \mid \text{model} \mid \text{nullable_by_default}$

— JML Syntax —

Table A.4 shows annotations that may appear inside method bodies, such as loop invariants, block contracts [Wacker, 2012], or ghost assignment statements. Set statements may appear whenever the Java language expects a statement [Gosling et al., 2013, Sect. 14.5], block contracts may appear immediately before a statement block, and loop invariants must appear immediately before a loop statement (**while**, (enhanced) **for**, or **do** loops; see *ibid.*, Sects. 14.12ff.). As a technical restriction of the implementation of KeY, any annotation statement must not be the last statement in a statement block.

Table A.4 JML annotation grammar

— JML Syntax —

$\langle Annot \rangle ::= \langle LoopInv \rangle \mid \langle BlockCntr \rangle \mid \langle SetStm \rangle \mid \langle Assert \rangle \mid \text{unreachable};$
 $\langle LoopInv \rangle ::= (\langle LoopInvClause \rangle)^+ (\langle VariantClause \rangle)^? (\langle Assign \rangle)^? (\langle Determs \rangle)^?$
 $\langle LoopInvClause \rangle ::= (\text{maintaining} \mid \text{loop_invariant}) \langle BoolExpr \rangle$
 $\langle VariantClause \rangle ::= (\text{decreasing} \mid \text{decreases}) \langle Exprs \rangle$
 $\langle BlockCntr \rangle ::= \text{also}^? \langle BSpecCase \rangle (\text{also} \langle BSpecCase \rangle)^*$
 $\langle BSpecCase \rangle ::= \langle BBehavior \rangle^? \langle BClause \rangle^*$
 $\langle BBehavior \rangle ::= \text{behavior} \mid \text{normal_behavior} \mid \text{exceptional_behavior} \mid$
 $\quad \text{break_behavior} \mid \text{continue_behavior} \mid \text{return_behavior}$
 $\langle BClause \rangle ::= \langle Breaks \rangle; \mid \langle Returns \rangle; \mid \langle Clause \rangle$
 $\langle Breaks \rangle ::= (\text{breaks} \mid \text{continues}) \langle Id \rangle^? \langle BoolExpr \rangle$
 $\langle Returns \rangle ::= \text{returns} \langle BoolExpr \rangle$
 $\langle SetStm \rangle ::= \text{set} \langle Loc \rangle = \langle Expr \rangle;$
 $\langle Assert \rangle ::= \text{assert} \langle BoolExpr \rangle;$

— JML Syntax —

Table A.5 shows the grammar for expressions in JML. This treatment is complete; all other Java expressions are not valid in JML and thus are rejected by KeY’s parser. Rule $\langle Literal \rangle$ refers to (integer) literals of any type or numeral system (see *ibid.*, Sect. 15.8.1).

Table A.5 JML expression grammar

— JML Syntax —

```

⟨Exprs⟩ ::= (⟨Expr⟩,)* ⟨Expr⟩
⟨Expr⟩ ::= ⟨BoolExpr⟩ | ⟨IntExpr⟩ | ⟨LocSetExpr⟩ | ⟨SeqExpr⟩ | this | null
⟨BoolExpr⟩ ::= !⟨BoolExpr⟩ | ⟨BoolExpr⟩⟨BinaryBoolOp⟩⟨BoolExpr⟩ |
  ⟨Expr⟩⟨EqOp⟩⟨Expr⟩ | ⟨IntExpr⟩⟨CompareOp⟩⟨IntExpr⟩ |
  ⟨⟨Quant⟩ ⟨Type⟩ ⟨Id⟩+; (⟨BoolExpr⟩;)? ⟨BoolExpr⟩ |
  \invariant_for(⟨Expr⟩) | \static_invariant_for(⟨Type⟩) |
  \fresh(⟨Expr⟩) | \nonnullelements(⟨Expr⟩) |
  \is_initialized(\type(⟨Type⟩)) |
  \reach(⟨Id⟩, ⟨Expr⟩, ⟨Expr⟩(, ⟨IntExpr⟩)?) | ⟨Expr⟩ instanceof ⟨Type⟩ |
  \typeof(⟨Expr⟩) ⟨TypeOp⟩ \type(⟨Type⟩) | \subset(⟨SeqExpr⟩, ⟨SeqExpr⟩) |
  \disjoint(⟨SeqExpr⟩ (, ⟨SeqExpr⟩)+) | \new_elems_fresh(⟨LocSetExpr⟩) |
  true | false | ⟨GenExpr⟩
⟨BinaryBoolOp⟩ ::= && | & | || | | ==> | <== | <==> | <!=> | ^
⟨EqOp⟩ ::= == | !=
⟨CompareOp⟩ ::= < | <= | > | >=
⟨TypeOp⟩ ::= == | <:
⟨Quant⟩ ::= \forall | \exists
⟨IntExpr⟩ ::= -⟨IntExpr⟩ | ~⟨IntExpr⟩ | ⟨IntExpr⟩⟨BinaryIntOp⟩⟨IntExpr⟩ |
  ⟨IntExpr⟩⟨BinaryIntOpBitw⟩⟨IntExpr⟩ | ⟨Comprehension⟩ | \index |
  ⟨Expr⟩.length | ⟨Literal⟩ | ⟨GenExpr⟩
⟨BinaryIntOp⟩ ::= + | - | * | / | %
⟨BinaryIntOpBitw⟩ ::= << | >> | >>> | & | | | ^
⟨Comprehension⟩ ::= (⟨ComprOp⟩ ⟨Type⟩ ⟨Id⟩+; (⟨BoolExpr⟩;)? ⟨IntExpr⟩) |
  (\num_of ⟨Type⟩ ⟨Id⟩+; (⟨BoolExpr⟩;)? ⟨IntExpr⟩)
⟨ComprOp⟩ ::= \sum | \product | \max | \min
⟨GenExpr⟩ ::= ⟨BoolExpr⟩? ⟨Expr⟩: ⟨Expr⟩ | \result | \old(⟨Expr⟩) |
  \pre(⟨Expr⟩) | (\lblneg ⟨Id⟩ ⟨Expr⟩) | (\lblpos ⟨Id⟩ ⟨Expr⟩) |
  \dl_⟨Id⟩(⟨Exprs⟩?) | \exception |
  ⟨Type⟩ ⟨Expr⟩ | ⟨Loc⟩ | (⟨Expr⟩.)?⟨Id⟩(⟨Exprs⟩?) |
  ⟨Type⟩ ⟨SeqExpr⟩[⟨IntExpr⟩] | (* ⟨JavaDLTerm⟩ *)

⟨Loc⟩ ::= (⟨Expr⟩.)?⟨Id⟩ | ⟨Type⟩.⟨Id⟩ | ⟨Expr⟩[⟨IntExpr⟩]
⟨LocSetExpr⟩ ::= ⟨Expr⟩.⟨Id⟩ | ⟨Expr⟩[⟨IntExpr⟩] | ⟨Expr⟩[⟨IntExpr⟩..⟨IntExpr⟩] |
  ⟨Expr⟩[*] | ⟨Expr⟩.* | \empty | \everything |
  ⟨LocSetOp⟩(⟨LocSetExpr⟩, ⟨LocSetExpr⟩) |
  \infinite_union(⟨Type⟩ ⟨Id⟩; (⟨BoolExpr⟩; )?⟨LocSetExpr⟩) |
  \reachLocs(⟨Id⟩, ⟨Expr⟩(, ⟨IntExpr⟩)?) | ⟨GenExpr⟩
⟨LocSetOp⟩ ::= \intersect | \set_union | \set_minus
⟨SeqExpr⟩ ::= \seq_empty | \seq_singleton(⟨Expr⟩) | \values |
  \seq_concat(⟨SeqExpr⟩, ⟨SeqExpr⟩) |
  ⟨SeqExpr⟩[⟨IntExpr⟩..⟨IntExpr⟩] | ⟨GenExpr⟩
  (\seq_def ⟨Type⟩ ⟨Id⟩; ⟨IntExpr⟩; ⟨IntExpr⟩; ⟨Expr⟩)
⟨Type⟩ ::= boolean | byte | char | short | int | long | \bigint | \seq |
  \locset | ⟨NullMod⟩? ⟨Id⟩(□)*
⟨NullMod⟩ ::= nullable | non_null

```

— JML Syntax —

A.2 JML Expression Semantics

This section contains semantics for JML expressions, given as translations into JavaDL. Tables A.6–A.8 contain translations for Boolean expressions (yielding formulas). For some of the logical operators two or more alternatives with the same semantics exist.¹ All other tables contain translations to terms. See Section 8.1 for details.

Table A.6 Translation of Boolean JML operators

$e \in \text{JExp}$ (alternatives)	JavaDL $[e] \in \text{DLFml}$
$\neg A$	$\neg[A]$
$A \ \&\& \ B$ $A \ \& \ B$	$[A] \wedge [B]$
$A \ \ B$ $A \ \ B$	$[A] \vee [B]$
$A \ ==> \ B$ $B \ <== \ A$	$[A] \rightarrow [B]$
$A \ <==> \ B$ $A \ == \ B$	$[A] \leftrightarrow [B]$
$A \ <!=> \ B$ $A \ \sim \ B$ $A \ != \ B$	$\neg([A] \leftrightarrow [B])$
$(\backslash\text{forall } T \ x_1, \dots, x_n; A; B)$	$\forall [T]x_1; \dots \forall [T]x_n; (\bigwedge_{i=1}^n \text{inRange}_T(x_i) \wedge [A] \rightarrow [B])$
$(\backslash\text{exists } T \ x_1, \dots, x_n; A; B)$	$\exists [T]x_1; \dots \exists [T]x_n; (\bigwedge_{i=1}^n \text{inRange}_T(x_i) \wedge [A] \wedge [B])$

Table A.7 Translation of special Boolean JML operators

JML expression e	JavaDL $[e] \in \text{DLFml}$
$\backslash\text{invariant_for}(o)$	$\text{Object}::\text{inv}(\text{heap}, [o])$
$\backslash\text{static_invariant_for}(T)$	$[T]::\text{\$inv}(\text{heap})$
$\backslash\text{fresh}(o)$	$\text{select}_{\text{boolean}}(\text{heap}^{\text{pre}}, [o], \text{created}) \doteq \text{FALSE} \wedge [o] \neq \text{null}$
$\backslash\text{nonnullelements}(a)$	$[a] \neq \text{null} \wedge \forall i. (0 \leq i < [a].\text{length} \rightarrow \text{select}_{\text{Object}}(\text{heap}, [a], \text{arr}(i)) \neq \text{null})$
$\backslash\text{is_initialized}(\backslash\text{type}(T))$	$[T].\text{<classInitialised>} \doteq \text{TRUE}$
$\backslash\text{reach}(f, o_1, o_2, n)$	$\text{reach}(\text{heap}, \text{allObjects}(f), [o_1], [o_2], [n])$
$\backslash\text{reach}(f, o_1, o_2)$	$\exists \text{int } n; \text{reach}(\text{heap}, \text{allObjects}(f), [o_1], [o_2], n)$
$\backslash\text{typeof}(x) == \backslash\text{type}(T)$	$\text{exactInstance}_{[T]}([x])$
$x \ \text{instanceof} \ T$	$\text{instance}_{[T]}([x])$

Table A.8 Predicates on location sets

JML expression e	JavaDL $[e] \in \text{DLFml}$
$\backslash\text{subset}(s, t)$	$\text{subset}([s], [t])$
$\backslash\text{disjoint}(s_1, \dots, s_n)$	$\bigwedge_{1 \leq i < j \leq n} \text{disjoint}([s_i], [s_j])$
$\backslash\text{fresh}(s)$	$\text{subset}([s], \text{unusedLocs}(\text{heap}^{\text{pre}}))$
$\backslash\text{new_elems_fresh}(s)$	$\text{subset}([s], \text{union}(\{\text{heap} := \text{heap}^{\text{pre}}\}[s], \text{unusedLocs}(\text{heap}^{\text{pre}})))$

Table A.9 Translation of Java integer expressions. Depending on the type of subexpressions, there are different translations: The right-most column applies if at least one operand is of type `\bigint`, else if at least one operand is of type `long`, the center column applies, and else the left column applies. Note that there is no cast to `\bigint`. Refer to Section 5.4 for a detailed explanation on these functions.

$e \in \text{JExp}$	$ e \in \text{Trm}_{int}$ $\text{type}_{\text{JML}}(e) = \text{int}$	$ e \in \text{Trm}_{int}$ $\text{type}_{\text{JML}}(e) = \text{long}$	$ e \in \text{Trm}_{int}$ $\text{type}_{\text{JML}}(e) = \backslash\text{bigint}$
$- n$	<code>javaUnaryMinusInt(n)</code>	<code>javaUnaryMinusLong(n)</code>	$-[n]$
$n + m$	<code>javaAddInt(n, m)</code>	<code>javaAddLong(n, m)</code>	$[n] + [m]$
$n - m$	<code>javaSubInt(n, m)</code>	<code>javaSubLong(n, m)</code>	$[n] - [m]$
$n * m$	<code>javaMulInt(n, m)</code>	<code>javaMulLong(n, m)</code>	$[n] * [m]$
n / m	<code>javaDivInt(n, m)</code>	<code>javaDivLong(n, m)</code>	<code>div(n, m)</code>
$n \% m$	<code>javaModInt(n, m)</code>	<code>javaModLong(n, m)</code>	<code>mod(n, m)</code>
$(T) n$	<code>castToByte(n), <code>castToShort(n), <code>castToLong(n), or <code>castToChar(n)</code></code></code></code>		
$\sim n$	<code>javaBitwiseNegation(n)</code>		—
$n \ll m$	<code>javaShiftLeftInt(n, m)</code>	<code>javaShiftLeftLong(n, m)</code>	—
$n \gg m$	<code>javaShiftRightInt(n, m)</code>	<code>javaShiftRightLong(n, m)</code>	—
$n \ggg m$	<code>javaUnsignedShiftRightInt(n, m)</code>	<code>javaUnsignedShiftRightLong(n, m)</code>	—
$n \& m$	<code>javaBitwiseAndInt(n, m)</code>	<code>javaBitwiseAndLong(n, m)</code>	—
$n m$	<code>javaBitwiseOrInt(n, m)</code>	<code>javaBitwiseOrLong(n, m)</code>	—
$n \sim m$	<code>javaBitwiseXorInt(n, m)</code>	<code>javaBitwiseXorLong(n, m)</code>	—

Table A.10 Translation of comprehension expressions (generalized quantifiers)

JML expression e	JavaDL $[e] \in \text{Trm}_{int}$	$[T]$
$(\backslash\text{sum } T \ x; n \leq x \ \&\& \ x < m; t)$	$(T')b\text{sum}\{int\ x\}(\lfloor n \rfloor, \lfloor m \rfloor, \lfloor t \rfloor)$	$[T] = int, t \text{ of type } T'$
$(\backslash\text{sum } T \ x_1, \dots, x_n; A; t)$	$(T')s\text{um}\{T \ x_1\}(\dots, \text{sum}\{T \ x_n\}(\bigwedge_{i=1}^n \text{inRange}_T(x_i) \wedge [A], \lfloor t \rfloor) \dots)$	$t \text{ of type } T'$
$(\backslash\text{num_of } T \ x; n \leq x \ \&\& \ x < m; B)$	$(T')b\text{sum}\{int\ x\}(\lfloor n \rfloor, \lfloor m \rfloor, \text{if } (\lfloor B \rfloor) \text{ then } (1) \text{ else } (0))$	$[T] = int, t \text{ of type } T'$
$(\backslash\text{num_of } T \ x_1, \dots, x_n; A; B)$	$(T')s\text{um}\{T \ x_1\}(\dots, \text{sum}\{T \ x_n\}(\bigwedge_{i=1}^n \text{inRange}_T(x_i) \wedge [A], \text{if } (\lfloor B \rfloor) \text{ then } (1) \text{ else } (0)) \dots)$	$t \text{ of type } T'$
$(\backslash\text{product } T \ x; n \leq x \ \&\& \ x < m; t)$	$(T')p\text{rod}\{int\ x\}(\lfloor n \rfloor, \lfloor m \rfloor, \lfloor t \rfloor)$	$[T] = int, t \text{ of type } T'$
$(\backslash\text{product } T \ x_1, \dots, x_n; A; t)$	$(T')p\text{rod}\{T \ x_1\}(\dots, \text{prod}\{T \ x_n\}(\bigwedge_{i=1}^n \text{inRange}_T(x_i) \wedge [A], \lfloor t \rfloor) \dots)$	$t \text{ of type } T'$
$(\backslash\text{max } T \ x_1, \dots, x_n; A; t)$	$\text{max}\{T \ x_1\}(\dots, \text{max}\{T \ x_n\}(\bigwedge_{i=1}^n \text{inRange}_T(x_i) \wedge [A], \lfloor t \rfloor) \dots)$	$t \text{ of type } T'$
$(\backslash\text{min } T \ x_1, \dots, x_n; A; t)$	$\text{min}\{T \ x_1\}(\dots, \text{min}\{T \ x_n\}(\bigwedge_{i=1}^n \text{inRange}_T(x_i) \wedge [A], \lfloor t \rfloor) \dots)$	$t \text{ of type } T'$

Table A.11 Restrictions on JavaDL types to match JML types

JML Type T'	JavaDL type $[T']$	Restriction formula $\text{inRange}_{T'}(x)$
boolean	boolean	true
byte	int	$\text{inByte}(x)$
char	int	$\text{inChar}(x)$
short	int	$\text{inShort}(x)$
int	int	$\text{inInt}(x)$
long	int	$\text{inLong}(x)$
\bigint	int	true
nullable T	$T \sqsubseteq \text{Object}$	$\text{select}_{\text{boolean}}(\text{heap}, x, \text{created}) \doteq \text{TRUE} \vee x \doteq \text{null}$
nonnull T	$T \sqsubseteq \text{Object}$	$\text{select}_{\text{boolean}}(\text{heap}, x, \text{created}) \doteq \text{TRUE} \wedge x \neq \text{null}$
nonnull $T \square^n$	$T \sqsubseteq \text{Object}$	$\text{select}_{\text{boolean}}(\text{heap}, x, \text{created}) \doteq \text{TRUE} \wedge \text{nonNull}(\text{heap}, x, n)$
\locset	LocSet	$\text{disjoint}(x, \text{unusedLocs}(\text{heap}))$
\seq	Seq	true

Table A.12 Translation of special JML operators of arbitrary type

JML expression e	JavaDL $[e] \in \text{Trm}_{Any}$
$A? B: C$	if ($[A]$) then ($[B]$) else ($[C]$)
$\backslash\text{result}$	res
$\backslash\text{old}(A)$	$\{\text{heap} := \text{heap}^{pre} \parallel p_1 := p_1^{pre} \parallel \dots \parallel p_n := p_n^{pre}\} [A]$ (in loop invs.)
$\backslash\text{old}(A)$	$\{\text{heap} := \text{heap}^{pre}\} [A]$ (in method contracts)
$\backslash\text{pre}(A)$	$\{\text{heap} := \text{heap}^{pre}\} [A]$
$(\backslash\text{blneg } x A)$	$[A]$
$(\backslash\text{blpos } x A)$	$[A]$
$\backslash\text{dl_func}(p_1, \dots, p_n)$	$\text{func}(\text{heap}, [p_1], \dots, [p_n])$
$(* \text{term} *)$	term
$\backslash\text{index}$	index

Table A.13 Reference expressions

	$e \in \text{JExp}$	$[e] \in \text{Trm}_{Any}$
self reference	this	self
local variable	v	v
field access	$o.f$	$\text{select}_{T'}(\text{heap}, [o], C::f)$
static field access	$C.f$	$\text{select}_{T'}(\text{heap}, \text{null}, C::\$f)$
array access	$a[i]$	$\text{select}_{T'}(\text{heap}, [a], \text{arr}([i]))$
array length	$a.\text{length}$	$\text{length}([a])$
pure method	$o.\text{pm}(p_1, \dots, p_n)$	$C::\text{pm}(\text{heap}, [o], [p_1], \dots, [p_n])$
static pure method	$C.\text{pm}(p_1, \dots, p_n)$	$C::\text{pm}(\text{heap}, \text{null}, [p_1], \dots, [p_n])$
model field	$o.\text{mf}$	$C::\text{mf}(\text{heap}, [o])$
static model field	$C.\text{mf}$	$C::\text{mf}(\text{heap}, \text{null})$

A.3 JML Expression Well-Definedness

As explained in Section 8.1.4, in JML, validity depends on the absence of undefinedness. According to the JML reference manual [Leavens et al., 2013], a Boolean expression is satisfied in a state if has the truth value true and “does not cause an exception to be raised.” Our translation from JML to JavaDL above ignores this.

KeY can generate well-definedness proof obligations as shown in Section 8.3.3. Table A.16 below gives the full definition of the well-definition operator ω , which provides a formula $\omega(e)$ to every JML expression e , such that e is well-behaving in a state s if and only if $s \models \omega(e)$.

¹ These operators may differ in the way well-definedness of expressions is evaluated, see Section A.3 below.

Table A.14 JML location set expressions

JML expression e	JavaDL $[e] \in \text{Trm}_{\text{LocSet}}$
$o.f$	$\{([o], f)\}$
$\backslash\text{singleton}(o.f)$	$\{([o], f)\}$
$a[i]$	$\{([a], \text{arr}([i]))\}$
$a[i..j]$	$\text{arrayRange}([a], [i], [j])$
$a[*]$	$\text{allFields}([a])$
$o.*$	$\text{allFields}([o])$
$\backslash\text{empty}$	empty
$\backslash\text{nothing}$	empty
$\backslash\text{everything}$	$\text{setMinus}(\text{allLocs}, \text{unusedLocs}(\text{heap}))$
$\backslash\text{intersect}(s, t)$	$\text{intersect}([s], [t])$
$\backslash\text{set_union}(s, t)$	$\text{union}([s], [t])$
$\backslash\text{set_minus}(s, t)$	$\text{setMinus}([s], [t])$
$\backslash\text{infinite_union}(T\ x; b; t)$	$\text{infiniteUnion}\{[T]\ x\}(\text{if } (\text{inRange}_T(x) \wedge [b]) \text{ then } ([t]) \text{ else } (\text{empty}))$
$\backslash\text{reachLocs}(f, o, n)$	$\text{infiniteUnion}\{\text{Object } o'\}(\text{if } (\text{reach}(\text{heap}, \text{allObjects}(f), [o], o', [n])) \text{ then } (\text{allFields}(o')) \text{ else } (\text{empty}))$
$\backslash\text{reachLocs}(f, o)$	$\text{infiniteUnion}\{\text{Object } o'\}(\text{if } (\exists \text{int } n; \text{reach}(\text{heap}, \text{allObjects}(f), [o], o', n)) \text{ then } (\text{allFields}(o')) \text{ else } (\text{empty}))$

Table A.15 JML sequence expressions

JML expression e	JavaDL $[e] \in \text{Trm}_{\text{Any}}$
$\backslash\text{seq_concat}(s1, s2)$	$\text{seqConcat}([s1], [s2])$
$\backslash\text{seq_empty}$	seqEmpty
$(T)s[i]$	$\text{seqGet}_T([s], [i])$
$s.\text{length}$	$\text{seqLen}([s])$
$\backslash\text{seq_singleton}(e)$	$\text{seqSingleton}([e])$
$s[i..j]$	$\text{seqSub}([s], [i], [j])$
$(\backslash\text{seq_def } \backslash\text{bigint } x; i; j; t)$	$\text{seqDef}\{\text{int } x\}([i], [j], [t])$
$\backslash\text{values}$	values

Table A.16 Definition of the well-definedness operator ω

JML expression $e \in \text{JExp}$	JavaDL formula $\omega(e) \in \text{DLFml}$	
x, this, super, null, \result	<i>true</i>	for x a literal or local variable
$\circ A$	$\omega(A)$	$\circ \in \{!, -, \sim\}$
$A \circ B$	$\omega(A) \wedge \omega(B)$	$\circ \in \{==, !=, <=, >=, <, >, +,$ $-, *, \&, , <==>, \hat{>>, <<, >>>\}$
$\backslash f(A)$	$\omega(A)$	$f \in \{\text{fresh}, \text{new_elems_fresh}\}$
$\backslash f(A, B)$	$\omega(A) \wedge \omega(B)$	$f \in \{\text{set_union}, \text{intersect},$ $\text{set_minus}, \text{subset}, \text{disjoint}\}$
$A / B, A \% B$	$\omega(A) \wedge \omega(B) \wedge B \neq 0$	
$A \&\& B$	$\omega(A) \wedge ([A] \rightarrow \omega(B))$	
$A \Rightarrow B$	$\omega(A) \wedge ([A] \rightarrow \omega(B))$	
$A B$	$\omega(A) \wedge (\neg[A] \rightarrow \omega(B))$	
$A <== B$	$\omega(A) \wedge (\neg[A] \rightarrow \omega(B))$	
$A ? B : C$	$\omega(A) \wedge ([A] \rightarrow \omega(B))$ $\wedge (\neg[A] \rightarrow \omega(C))$	
$o.f$	$\omega(o) \wedge o \neq \text{null}$	for an instance field f, also if used as location
$C.f$	<i>true</i>	for a static field C.f, also if used as location
$o.m(a_1, \dots, a_n)$	$\omega(o) \wedge o \neq \text{null} \wedge$ $\bigwedge_{i=1}^n \omega(a_i) \wedge$ $\text{pre}[a_i/p_i, o/\text{self}]$	
$\text{new } C(a_1, \dots, a_n)$	$\bigwedge_{i=1}^n \omega(a_i) \wedge$ $\text{pre}[a_i/p_i, o/\text{self}]$	
$a[i]$	$\omega(a) \wedge a \neq \text{null} \wedge \omega(i) \wedge$ $0 \leq i \wedge i < \text{length}(a)$	array access, sequence access also if used as location.
$a[i..j]$	$\omega(a) \wedge a \neq \text{null} \wedge$ $\omega(i) \wedge \omega(j) \wedge$ $0 \leq i \wedge i \leq j \wedge j < \text{length}(a)$	array range (location set)
$o[*], o.*$	$\omega(o) \wedge o \neq \text{null}$	
$o.\text{length}$	$\omega(o) \wedge o \neq \text{null}$	for an array o
$(\backslash Q \ T \ v; A; B)$	$\forall [T] v; (\omega(A) \wedge ([A] \rightarrow \omega(B)))$	$Q \in \{\text{forall}, \text{exists}, \text{min}, \text{max}$ $\text{infinte_union}, \text{sum}, \text{product}\}$
$(\backslash \text{seq_def } T \ v; A; B; C)$	$\forall \text{int } v; ([A] \leq v < [B] \rightarrow \omega(C))$ $\wedge \omega(A) \wedge \omega(B)$	$T \in \{\text{int}, \backslash \text{bigint}\}$
$(T)t$	$\omega(t) \wedge \text{instance}_{[T]}([t]) \doteq \text{TRUE}$	
$t \text{ instanceof } T$	$\omega(t)$	
$\backslash \text{old}(A)$	$\{\text{heap} := \text{heap}^{\text{pre}}\} \omega(A)$	
$\backslash \text{invariant_for}(A)$	$\omega(A) \wedge A \neq \text{null}$	